

X.2. SYNTAX OF THE XPLAIN DATA LANGUAGE

Help symbols are enclosed by < and >. The symbol : stands for 'can be' and the symbol | stands for 'or'. The symbol ε denotes the empty string. The symbol □ is used for marking the end of a number of alternatives. Standard keywords (as **base**, **type**, **init**, **get**, etc.) and special symbols (as ., (, ", [) are printed in bold typeface.

DATA DEFINITION

```

<definition list>:
    <definition command>
    |
    <definition list> <definition command>
    □

<definition command>:
    <definition> .
    |
    <deletion> .
    |
    <assignment> .
    |
    <comment>
    □

<definition>:
    <type definition>
    |
    <base definition>
    |
    <init definition>
    |
    <default definition>
    |
    <constant definition>
    □

<deletion>:
    <type deletion>
    |
    <general deletion>
    □

```

<assignment>:
 <constant> = (<expression>)
 □

<base definition>:
base <base name> <domain>
 |
base <base name> <domain> <domain constraint>
 □

<type definition>:
type <type name> <domain> = <attribute list>
 □

<init definition>:
init <initialisation>
 □

<default definition>:
init default <initialisation>
 □

<constant definition>:
constant <constant> <domain>
 □

<type deletion>:
purge <type_or_constant>
 □

<general deletion>:
purge <type name> **its** <attribute name>
 |
purge init <type name> **its** <attribute name>
 □

<domain constraint>:
 = <enumeration>
 |
 <pattern>
 |
 <trajectory>
 □

```

<enumeration>:
  <string enumeration>
  |
  <integer enumeration>
  □

<string enumeration>:
  <text>
  |
  <string enumeration> , <text>
  □

<integer enumeration>:
  <number>
  |
  <integer enumeration> , <number>
  □

<trajectory>:
  ( border .. border )
  □

<border>:
  * | <number>
  □

<pattern>:
  <pattern character>
  |
  <pattern character> <pattern>
  □

<pattern character>:
  ? | x | 9 | - | . | , | \ | / | : | ; | <space>
  □

<subject>:
  <type name>
  □

<plus_or_minus>:
  + | -
  □

```

<mult_or_div>:

* | / | %
□

<logical expression>:

<logical expression> **or** <logical term>
|
<logical term>
□

<logical term>:

<logical term> **and** <logical factor>
|
<logical factor>
□

<logical factor>:

<logical expression>
|
not <logical expression>
|
<property expression> <relation> <property expression>
|
not <property expression> <relation> <property expression>
|
(<logical expression>)
|
not (<logical expression>)
|
<logical value>
|
not <logical value>
□

<relation>:

< | <= | <> | > | >= | =
□

<logical value>:

true | **false**
□

<system variable>:

systemdate | **loginname**
□

```

<property expression>:
  <property expression> <plus_or_minus> <property term>
  |
  <property term>
  □

<property term>:
  <property term> <mult_or_div> <property factor>
  |
  <property factor>
  □

<property factor>:
  <property name>
  |
  - <property name>
  |
  <system variable>
  |
  combine ( <property expression> , <property expression> )
  |
  head ( <property expression> )
  |
  tail ( <property expression> )
  |
  integer ( <property expression> )
  |
  real ( <property expression> )
  |
  string ( <property expression> )
  |
  datef ( <property expression> )
  |
  newdate ( <property expression> , <property expression> )
  |
  timedif ( <property expression> , <property expression> )
  |
  isdate ( <property expression> )
  |
  yearf ( <property expression> )
  |
  monthf ( <property expression> )
  |
  dayf ( <property expression> )
  |

```

```

wdayf ( <property expression> )
|
( <property expression> )
|
- ( <property expression> )
|
<text>
|
<numeric>
□

<numeric>:
<number>
|
<real>
□

<property name>:
<attribute name>
|
<attribute name> its <property name>
□

<attribute name>:
<prefix> <name>
□

<constant_or_property>:
<attribute name>
|
<constant>
|
<attribute name> its <constant_or_property>
□

<initialisation>:
<type name> its <attribute name> = <init specification>
□

<init specification>:
<property expression>
|
if <condition> then <init specification> else <init specification>
|

```

case <selector> **of** <case list>
 □

<condition>:
 <property expression>
 □

<selector>:
 <property expression>
 □

<case list>:
 <element list> <default element>
 □

<element list>:
 <element> ;
 |
 <element list> <element> ;
 □

<element>:
 <label list> : <init specification>
 □

<default element>:
default : <init specification>
 □

domain:
 (**A** number)
 |
 (**B**)
 |
 (**I** number)
 |
 (**R** number , number)
 |
 (**D**)
 □

<type name>:
 <name>
 □

```
<base name>:  
  <name>  
  □  
  
<constant>:  
  <name>  
  □  
  
<type_or_constant>:  
  <name>  
  □  
  
<prefix>:  
  ε  
  |  
  <name>_  
  □  
  
<attribute list>:  
  <definition attribute>  
  |  
  <attribute list> , <definition attribute>  
  □  
  
<definition attribute>:  
  [ <attribute> ]  
  |  
  <attribute>  
  □  
  
<attribute>:  
  <type name>  
  |  
  <prefix> <type name>  
  □  
  
<name>:  
  <letter followed by at most 19 letters and/or digits>  
  □  
  
<text>:  
  " <collection of printable characters> "  
  □
```

<number>:

<numeric value possibly preceded by a sign>

□

<date>:

<valid calendar date: 8 digits without sign according interpretation jjjjmmdd>

<real>:

<real numeric possibly preceded by a sign>

□

<comment>:

comment text <sign for newline>

□

<comment text>:

<collection of printable characters>

□

<string function>:

combine | **head** | **tail**

□

<conversion function>:

integer | **real** | **string** | **datef**

□

<date function>:

newdate | **timedif** | **isdate** | **yearf** | **monthf** | **dayf** | **wdayf**

□

DATA MANIPULATION

```
<manipulation list>:  
  <command> .  
  |  
  <query> <command> .  
  |  
  <comment>  
  □
```

```
<command>:  
  <retrieval>  
  |  
  <modification>  
  |  
  newline  
  □
```

```
<modification>:  
  <update>  
  |  
  <delete>  
  |  
  <insert>  
  |  
  <cascade>  
  □
```

```
<insert>:  
  insert <subject> its <assignment list>  
  |  
  insert <name> * its <assignment list>  
  □
```

```
<assignment list>:  
  <assignment> , <assignment list>  
  |  
  <assignment>  
  □
```

```

<assignment>:
  <attribute name> = <assigned value>
  □

<assigned value>:
  <property expression>
  |
  ( <logical expression> )
  □

<cascade>:
  <name> its <attribute name> = <cascade specification>
  □

<cascade specification>:
  <expression> <predicate>
  |
  ( <logical expression> ) <predicate>
  |
  <selection expression> per <property name>
  |
  <selection expression> per <property name> , <property name>
  □

<deletion>:
  delete <subject> <predicate>
  □

<subject>:
  <name> <idstring>
  □

<idstring>:
  ε
  |
  <text>
  |
  <name>
  □

<update>:
  update <subject> its <assignment list> <predicate>
  □

```

```

<attribute name>:
  <prefix> <name>
  □

<retrieval>:
  <selection>
  |
  <extension>
  |
  <value>
  |
echo <text>
  □

<value>:
value <name> = <value definition>
  □

<value definition>:
  <property expression>
  |
  <value_selection expression>
  |
  <input>
  □

<input>:
input <domain>
  |
input <domain> <text>
  □

<domain>:
( A <numeric> )
  |
( B )
  |
( I <numeric> )
  |
( R <numeric>, <numeric> )
  |
( D )
  □

```

```

<value_selection expression>:
  <set function> <subject> <property> <predicate>
  □

<selection>:
  get <selection expression>
  |
  get <text> <selection expression>
  □

<extension>:
  extend <extension expression>
  □

<extension expression>:
  <name> with <extend attribute> = <extension definition>
  □

<extend attribute>:
  <name>
  |
  <name> <domain>
  □

<extension definition>:
  <property expression>
  |
  ( <logical expression> )
  |
  <selection expression> per <property name>
  □

<selection expression>:
  <set function> <subject> <property> <predicate>
  |
  <subject> <property> <predicate>
  □

<set function>:
  max | min | total | count | some | nil | any
  □

<string function>:
  combine | head | tail
  □

```

<conversion function>:

integer | **real** | **string** | **datef**
□

<date function>:

newdate | **timedif** | **isdate** | **yearf** | **monthf** | **dayf** | **wdayf**
□

<mathematical function>:

pow | **abs** | **sqrt** | **max** | **min** | **exp** | **ln** | **log10** | **sin** | **cos** | **tan**
| **asin** | **acos** | **atan** | **sinh** | **cosh** | **tanh** | **asinh** | **acosh** | **atanh**

<property>:

ε
|
its <property list>
□

<property list>:

<property expression>
|
<property expression> , <property list>
□

<predicate>:

ε
|
where <logical expression>
□

<logical expression>:

<logical expression> **or** <logical term>
|
<logical term>
□

<logical term>:

<logical term> **and** <logical factor>
|
<logical factor>
□

<logical factor>:

<logical expression>
|

```

not <logical expression>
|
<property expression> <relation> <property expression>
|
not <property expression> <relation> <property expression>
|
(logicalexpression)
|
not (logicalexpression)
|
<logical value>
|
not <logical value>
□

<relation>:
< | <= | <> | > | >= | =
□

<logical value>:
true | false
□

<property expression>:
<property expression> <plus_or_minus> <property term>
|
<property term>
□

<plus_or_minus>:
+ | -
□

<property term>:
<property term> <mult_or_div> <property factor>
|
<property factor>
□

<mult_or_div>:
* | / | %
□

<property factor>:
<property name>

```

```
|  
| - <property name>  
|  
pow ( <property expression> , <property expression> )  
|  
log ( <property expression> )  
|  
sqrt ( <property expression> )  
|  
max ( <property expression>, <property expression> )  
|  
min ( <property expression>, <property expression> )  
|  
ln ( <property expression> )  
|  
exp ( <property expression> )  
|  
sin ( <property expression> )  
|  
cos ( <property expression> )  
|  
tan ( <property expression> )  
|  
abs ( <property expression> )  
|  
asin ( <property expression> )  
|  
acos ( <property expression> )  
|  
atan ( <property expression> )  
|  
asinh ( <property expression> )  
|  
acosh ( <property expression> )  
|  
atanh ( <property expression> )  
|  
sinh ( <property expression> )  
|  
cosh ( <property expression> )  
|  
tanh ( <property expression> )  
|  
combine ( <property expression> , <property expression> )  
|
```

```

head ( <property expression> )
|
tail ( <property expression> )
|
integer ( <property expression> )
|
real ( <property expression> )
|
string ( <property expression> )
|
datef ( <property expression> )
|
newdate ( <property expression> , <property expression> )
|
timedif ( <property expression> , <property expression> )
|
isdate ( <property expression> )
|
yearf ( <property expression> )
|
monthf ( <property expression> )
|
dayf ( <property expression> )
|
wdayf ( <property expression> )
|
( <property expression> )
|
- ( <property expression> )
|
<text>
|
loginname
|
<numeric>
□

<numeric>:
  <number>
  |
  <real>
  |
systemdate
□

```

```
<property name>:  
    <attribute name>  
    |  
    <attribute name> its <property name>  
    □  
  
<prefix>:  
    ε  
    |  
    <name>_  
    □  
  
<name>:  
    <letter followed by at most 19 letters and/or digits>  
    □  
  
<text>:  
    " <collection of printable characters> "  
    □  
  
<number>:  
    <numeric value possibly preceded by a sign>  
    □  
  
<date>:  
    <valid calendar date: 8 digits without sign according interpretation jjjjmmdd>  
  
<real>:  
    <real numeric possibly preceded by a sign>  
    □  
  
<comment>:  
    # comment text <sign for newline>  
    □  
  
<comment text>:  
    <collection of printable characters>  
    □
```